# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=scap-security-guide
pkgver=0.1.60
pkgrel=1
pkgdesc='Security compliance content in SCAP, Bash, Ansible, and other formats.'
arch=('any')
groups=('blackarch' 'blackarch-automation' 'blackarch-defensive'
        'blackarch-scanner')
url='https://www.open-scap.org/security-policies/scap-security-guide/'
license=('BSD')
depends=()
makedepends=('git')
optdepends=('openscap' 'scap-workbench')
source=("https://github.com/ComplianceAsCode/content/releases/download/v$pkgver/$pkgname-$pkgver.zip")
sha512sums=('15ab7480fc329baf192bac8474795ec8c4409be40bd01359b13a3c65f511322275fcded1de6ab79b3b70e37c24fe4629e5188c3fea29ef6e6bf839ea46c10d9d')

pkgver() {
  git ls-remote --sort=v:refname -t --refs \
    'https://github.com/ComplianceAsCode/content' 'v*' |
    grep -v 'InitialDraft' | sed -e 's/.*v//' -e '$!d'
}

package() {
  cd "$pkgname-$pkgver"

  install -dm 755 "$pkgdir/usr/share/$pkgname/"

  cp -a ./* "$pkgdir/usr/share/$pkgname/"
}

